Legkisebb négyzetek módszere

Legkisebb négyzetek módszere

1. Tétel Az alapprobléma és a tétel

Adott n db különböző pont, $x_1 < x_2 < \ldots < x_n$ $(x_j \in \mathbb{R})$ és n db függvényérték, $f_1, f_2, \ldots, f_n$ $(f_j \in \mathbb{R})$. Keressünk egy olyan legfeljebb M-edfokú p polinomfüggvényt, melyre $\sum_{j=1}^{n} (p[x_j] - f_j)^2$ minimális.

Tétel. Pontosan egy olyan legfeljebb M-edfokú (M≤ n-1) polinom létezik, melyre a négyzetösszeg minimális.

1. Megjegyzés

A p polinom α_j együtthatóit végtelen sokféleképpen választhajuk meg. A cél az optimális (véges) együttható-sorozat megtalálása. Tehát egy többváltozós optimalizálási problémával van dolgunk.

Kalkulus, normálegyenletek

Legyen $p = \sum_{j=0}^{m} \alpha_j\, x^j$ Mivel $G(\alpha_0, \alpha_1, \ldots) = \sum_{j=1}^{n} (p[x_j] - f_j)^2$ minden α_j szerint parciálisan differenciálható, a probléma vizsgálható a kalkulus eszközeivel.

1. Példa

M=0

X={1,2,3}, F={2,5,10}

```
P0[x_] := α0;
```

```
X = {1, 2, 3};
F = {2, 5, 10};
```

```
G = Plus @@ Map[(P0[#[[1]]] - #[[2]])^2 &, Transpose[{X, F}]]
```

$(-10 + \alpha 0)^2 + (-5 + \alpha 0)^2 + (-2 + \alpha 0)^2$

```
D[G, α0]
```

$2\ (-10+\alpha 0)+2\ (-5+\alpha 0)+2\ (-2+\alpha 0)$

```
Solve[D[G, α0] == 0]
```

$\left\{\left\{\alpha 0\to \frac{17}{3}\right\}\right\}$

```
Graphics[
 {RGBColor[1, 0, 0], PointSize[.03], Point[{1, 2}], Point[{2, 5}], Point[{3, 10}],
  RGBColor[0, 1, 0], Line[{{0, 17 / 3}, {4, 17 / 3}}]}, AspectRatio → .8]
```



2. PÉLDA

M=1

```
P1[x_] := α1 x + α0;
```

```
G = Plus @@ Map[(P1[#[[1]]] - #[[2]])^2 &, Transpose[{X, F}]]
```

$(-2+\alpha 0+\alpha 1)^2+(-5+\alpha 0+2\ \alpha 1)^2+(-10+\alpha 0+3\ \alpha 1)^2$

```
{D[G, α0], D[G, α1]}
```

$\{2\ (-2+\alpha 0+\alpha 1)+2\ (-5+\alpha 0+2\ \alpha 1)+2\ (-10+\alpha 0+3\ \alpha 1),$
$\ 2\ (-2+\alpha 0+\alpha 1)+4\ (-5+\alpha 0+2\ \alpha 1)+6\ (-10+\alpha 0+3\ \alpha 1)\}$

```
Solve[{D[G, α0] == 0, D[G, α1] == 0}]
```

$\left\{\left\{\alpha 0 \to -\frac{7}{3},\ \alpha 1 \to 4\right\}\right\}$

A kapott lin. polinom a 4x-7/ 3!

```
Graphics[
 {RGBColor[1, 0, 0], PointSize[.03], Point[{1, 2}], Point[{2, 5}], Point[{3, 10}],
  RGBColor[0, 1, 0], Line[{{0, -7 / 3}, {4, 41 / 3}}]}, AspectRatio → .8]
```



3. PÉLDA

M=2

```
P2[x_] := α2 x^2 + α1 x + α0;
```

```
G = Plus @@ Map[(P2[#[[1]]] - #[[2]])^2 &, Transpose[{X, F}]]
```

$(-2+\alpha 0+\alpha 1+\alpha 2)^2+(-5+\alpha 0+2\,\alpha 1+4\,\alpha 2)^2+(-10+\alpha 0+3\,\alpha 1+9\,\alpha 2)^2$

```
{D[G, α0], D[G, α1], D[G, α2]}
```

$\{2\,(-2+\alpha 0+\alpha 1+\alpha 2)+2\,(-5+\alpha 0+2\,\alpha 1+4\,\alpha 2)+2\,(-10+\alpha 0+3\,\alpha 1+9\,\alpha 2),$
$2\,(-2+\alpha 0+\alpha 1+\alpha 2)+4\,(-5+\alpha 0+2\,\alpha 1+4\,\alpha 2)+6\,(-10+\alpha 0+3\,\alpha 1+9\,\alpha 2),$
$2\,(-2+\alpha 0+\alpha 1+\alpha 2)+8\,(-5+\alpha 0+2\,\alpha 1+4\,\alpha 2)+18\,(-10+\alpha 0+3\,\alpha 1+9\,\alpha 2)\}$

```
Solve[{D[G, α0] == 0, D[G, α1] == 0, D[G, α2] == 0}]
{{α0 → 1, α1 → 0, α2 → 1}}
```

A kapott kvadratikus polinom az $x^2 + 1$!

```
InterpolatingPolynomial[Transpose[{X, F}], x] // Expand
1 + x^2
```

```
Plot[Evaluate[InterpolatingPolynomial[Transpose[{X, F}], x]],
 {x, 0, 4}, ImageSize → {300, 300}, PlotStyle → {RGBColor[0, 1, 0]},
 Epilog → {RGBColor[1, 0, 0], PointSize[.03], Map[Point[#] &, Transpose[{X, F}]]},
 PlotRange → {0, 20}]
```



A LS min. probléma felfogahtó az interp. probléma általánosításaként!

Ortogonalitás (v. diszkrét pontrendszer)

1. DEFINÍCIÓ

$$< q_j \mid q_l > = \sum_i q_j[x_i]\, q_l[x_i]$$

2. DEFINÍCIÓ

$\{q_0, q_1, \ldots\}$ ortogonális polinomrendszer (OPR), ha $j \neq l \Longrightarrow < q_j \,|q_l > = 0$

4. PÉLDA

X={1,2,3}

q0=1, q1=x-2, q2= x^2-4x+10/ 3

```
X = {1, 2, 3};
```

```
IP[f1_, f2_, var_] := Sum[(f1 /. var → X[[i]]) (f2 /. var → X[[i]]), {i, 1, Length[X]}]
```

```
IP[x - 2, x^2 - 4 x + 10 / 3, x]
0
```

Hogyan jön ki? Hint: $x^m + \sum \lambda_j \, q_j \quad \lambda_l = - < x^m \,|q_l > / < q_l \,|q_l >$

Megoldás Ortogonális polinomrendszer (OPR) felhasználásával

2. TÉTEL

Ötlet: $p = \sum \alpha_j \; q_j$, tehát keressük ortogonális oplinomrendsze kombinációjaként a megoldást!Kérdés: mik lesznek az α együtthatók?

$\alpha_j = < f \,|q_j > / < q_j \,|q_j >$

5. PÉLDA

```
X = {1, 2, 3};
F = {2, 5, 10}
```

OPR

```
Q = {1, x - 2,  x^2 - 4 x + 10 / 3 };
```

```
OptCoeff[OPR_, ind_] :=
 Module[{v = Table[OPR[[ind + 1]] /. x → X[[j]], {j, Length[OPR]}]}, v.F / v.v]
```

M=0

$< f \, | q_0 > / < q_0 | q_0 >$

```
{2, 5, 10}.{1, 1, 1} / {1, 1, 1}.{1, 1, 1}
```

$\frac{17}{3}$

```
OptCoeff[Q, 0]
```

$\frac{17}{3}$

M=1

$< f \, | q_1 > / < q_1 | q_1 >, \; < f \, | q_0 > / < q_0 | q_0 >$

```
{2, 5, 10}.{-1, 0, 1} / {-1, 0, 1}.{-1, 0, 1}
```

4

```
4 (x - 2) + 17 / 3 // Expand
```

$-\frac{7}{3} + 4 x$

```
OptCoeff[Q, 1]
```

4

M=2

```
{2, 5, 10}.{1 / 3, -2 / 3, 1 / 3} / {1 / 3, -2 / 3, 1 / 3}.{1 / 3, -2 / 3, 1 / 3}
```

1

```
OptCoeff[Q, 2]
```

1

```
1 (x^2 - 4 x + 10 / 3) + 4 (x - 2) + 17 / 3 // Expand
```

$1 + x^2$

$\sum_{j=0}^{2}$ `OptCoeff[Q, j] Q[[j + 1]] // Expand`

$1 + x^2$

Mit veszünk észre, ha összevetjük a kalkulusból származtatott megoldással?

Válasz: ugyanazt a megoldást kapjuk.

* Diszkrét Parseval formula

$$\| f \|^2 = \sum \left(< f , q_j > / \| q_j \| \right)^2$$

```
Norm[{2, 5, 10}]^2
```

129

```
({2, 5, 10}.{1, 1, 1} / Norm[{1, 1, 1}])^2 +
 ({2, 5, 10}.{-1, 0, 1} / Norm[{-1, 0, 1}])^2 +
 ({2, 5, 10}.{1 / 3, -2 / 3, 1 / 3} / Norm[{1 / 3, -2 / 3, 1 / 3}])^2
```

129

```
Norm[F]^2
```

129

$$\sum_{j=0}^{2} \texttt{OptCoeff}[Q, j]^2 \; \texttt{IP}[Q[[j+1]], Q[[j+1]], x]$$

129